
Summary

The goal of this two-part paper is to get you up to speed automating Excel and Word from a Delphi application. Most of the text will focus on Excel, but you will find that if you understand Excel automation, that you need only a few hints to get started automating Word. If you are primarily interested in Word, I ask you to patiently read through the material on Excel, as almost all of it applies to Word. 

You can use Delphi to fully control virtually all the features of Excel and Word. There is very little that you can do from inside Excel or Word that you cannot also automate from outside Excel or Word. In other words, both Excel and Word can be fully controlled from Delphi applications using OLE Automation. 

For the most part the act of controlling Excel or Word from a Delphi application is not terribly challenging. Whatever difficulty there is comes not from Delphi's side of the equation, but from the innate complexity of the Excel and Word object hierarchies. Not that I find the hierarchies unreasonable, but they do encapsulate sufficient complexity to require a significant period of study. In particular, these automation classes give you rather detailed control over Word and Excel. Since these are complex applications, the interface to them also needs to be complex if it is not to be considered under powered. The purpose of this paper is to unscramble that hierarchy and show its underlying structure. 

This paper is divided into two parts. The first part shows automating Word and Excel using variants and IDispatch, and the second part shows how to perform the same tasks with dispinterfaces and standard COM interfaces. All of these technologies are closely related. However, there are two significant differences between them: 

1. Using variants is usually the easiest, the most terse, but also the slowest way to get things done in terms of performance. 

2. Using COM interfaces is usually the most difficult, the most verbose, but also yields the highest performance. 

There may be a short gap between the time the first and second parts of the paper are published. 

When writing this paper I have attempted to explain things as clearly as possible. I'm aiming this paper primarily at intermediate or experienced programmers, but I hope it is accessible to anyone who has a basic understanding of how to use Delphi, Word and Excel. Though you should not need a high level of expertise to understand this paper, I am trying to cover the subject in some depth. Other sources, such as my book Dephi 2 Unleashed (Chapter 29), and the magazine the Delphi Informant (June 1997), covers some of this same material in a simpler, less in depth fashion. For many people a more high level, abstracted view may be more appropriate. But I believe there is also a big need for a more detailed look at this subject, which is why I have written this paper. 

One final note: in my writing, I tend to use repetition to emphasize important points. This is a consciously exercised technique designed to make this paper as easy as possible to understand, while simultaneously ensuring that you don't accidentally skip over an essential point. Another benefit of repetition is that people tend to use articles of this type as a reference. As a result, they don't always read them from beginning to end every time they refer to them. As a result, it helps to give a one sentence review of key points that relate to the topic under discussion. 

System Requirements
This paper was written against Delphi 3.01 and Microsoft Office 97. Portions of the paper would also work with Office 95, but the sections on interfaces, in particular, require that you use Office 97. 

To perform automation successfully with Excel or Word you need a fairly powerful system with lots of RAM. I've been automating Excel for at least four years. When I first started out, I considered the technology a bit suspect simply because it was terribly slow. Now, however, our machines are powerful enough to take Excel through its paces in a few short moments. In particular, if you have a Pentium 120 class machine or above, and at least 48 MB of ram, then this technology works well for many types of projects. Excel or Word will now load quite quickly, and you can open and insert data into them in the blink of an eye. However, if you want to iterate over lots of data inside a Word or Excel document, then that can be a bit time consuming when compared to performing similar tasks inside of a Delphi application. 

The bottom line here is that if you know Excel can do something well, and you know your target machines are powerful and will have Excel loaded on them, then there is no reason to search for third party components to perform spreadsheet related functions. Instead, you can just automate Excel from inside a Delphi application and get your work done professionally in just a few short hours. The icing on the cake is that you can then use MAPI to mail the results of your work to anyone who has a mail system and the ability to read Excel files. The point being that the recipient of your work need not actually have a copy of your Delphi application running when viewing the output from your program. Instead you can just send them the results in an Excel or Word document. Word document viewers can be downloaded for free from Microsoft's web site at www.microsoft.com. 

Getting Started with Delphi and Excel
There are two different ways to run OLE automation from inside Delphi. One involves using interfaces, while the second involves using an OLE class called IDispatch along with a Delphi type called a variant. Interfaces give you the advantage of type checking your code on the client side, as well as relatively high performance. However, I am going to start the paper working with the somewhat easier to understand IDispatch and variants technology, and move on to cover interfaces after all the basics are clearly established. Do not worry if you don't yet understand the differences between the two techniques, as this subject will be cleared up over the course of the paper. At this stage, you just need to be aware that there are at least two ways to access OLE automation objects from Delphi, and that I am going to start out by showing you one that uses IDispatch and variants. 

The following code, found in Listing 1, shows a bare bones example of a Delphi application that launches Excel. Just skim over the code for now, as I will spend the rest of this section of the paper explaining how it works. 

Listing 1: The main form from the Excel1 application found with the code samples that accompany this article. 

unit Main;

interface

uses

  Windows, Messages, SysUtils,

  Classes, Graphics, Controls,

  Forms, Dialogs, StdCtrls;

type

  TForm1 = class(TForm)

    Button1: TButton;

    procedure Button1Click(Sender: TObject);

    procedure FormDestroy(Sender: TObject);

  private

  public

    V: Variant;

  end;

var

  Form1: TForm1;

implementation

uses

  ComObj;

{$R *.DFM}

procedure TForm1.Button1Click(Sender: TObject);

begin

  V := CreateOleObject('Excel.Application');

  V.Visible := True;

end;

procedure TForm1.FormDestroy(Sender: TObject);

begin

  if not VarIsEmpty(V) then

    V.Quit;

end;

end.

You can find this example on disk, in the program called Excel1.dpr. The code does nothing more than create an instance of Excel, make it visible, and then close it down when the user exists the Delphi application. The code does not check to make sure the user is not creating multiple instances of the application, but it does close down a single copy of Excel when you exit. 

From Delphi's side, you should always start your automation applications by including COMObj in the uses clause. COMObj contains routines for retrieving OLE automation objects, and for dispatching calls to them. In particular, you usually use CreateOleObject to retrieve an automation object, and behind the scenes Delphi uses the COMObj routines VarDispInvoke, DispatchInvoke and GetIDsOfNames, to call an object. I will give you a bit more information on these routines in just one moment. The rest is simply a matter of using a built in Delphi type called a variant to reference the objects that reside inside Excel. 

Here are a few simple lines of code that launch Excel from inside Delphi: 

var

  V: Variant;

begin

  V := CreateOleObject('Excel.Application');

  V.Visible := True;

end;

The first line of code after the begin statement launches Excel. The call to CreateOleObject is relatively complex, so I will explain it in the next section of the paper. 

After executing the first line, Excel will come up in the background, entirely offscreen, invisible to the user. This may, in fact, be the effect you want to achieve. However, when you first start out programming Excel, and whenever you are debugging your Excel automation application, you probably want to be able to see what is going on inside Excel. Therefore, I set the Visible property of the Excel Application object equal to True. This ensures that you can see what is actually happening on the Excel server. If you have thoroughly debugged your application you may want to skip this step, but I will include it in all the examples.zip I cover in this paper. 

Of course, I haven't told you enough yet to make the code shown above entirely comprehensible. What is the purpose, for instance, of the variable V? What does CreateOleObject actually do? 

As it turns out, the answers to these questions are non-trivial. The variable V is a variant, and CreateOleObject creates an instance of a COM object called IDispatch and returns it to you inside a variant. But saying as much doesn't help much if you don't understand COM, IDispatch, and variants. 

There are three courses I could take at this point. One would involve an in depth explanation of COM and OLE, a second would give you only the minimal amount of information needed to keep going, and the third would be to find some middle ground. In this case I am going to opt for the latter solution, and leave it up to you to pursue COM in more depth on your own. I will however, discuss this subject over the next few paragraphs, and come back to it again in the second part of the paper. If you want even more information, some references to get you started are as follows: 

· I have an article on Delphi and DCOM on my WEB site: http://users.aol.com/charliecal, and also cover the subject in the book Delphi 2 Unleashed. 

· Microsoft has extensive documentation on this subject: http://ww.microsoft.com/com 

· The User's Guide from the Delphi 3 docs contains some text on this subject in Chapter 25. 

The only thing you really need to know at this stage is that Microsoft has created a special type of object oriented programming called COM, which allows you to retrieve and call the methods of an object from a number of different languages. The COM object model is different from the one used by native Delphi programmers, so Borland gives you two choices: 

You can followed Microsoft's lead and call the methods of these special objects off a variable type called a variant. This is the technique described in the first part of this paper. 

You can follow a second, more technical approach and use interfaces or dispinterfaces. 

One of the key differences between using interfaces and using variants is that interfaces allow you to call COM objects using the much faster dispatching technologies native to Object Pascal. As I will explain in Part II of this article, dispinterfaces follow a middle path between the variant technology and the interface technology. 

COM is the underlying object model that makes OLE and ActiveX programming possible. At times I will use OLE and COM as virtual synonyms. ActiveX is yet a third very closely related technology, but I will not touch on it in this paper. However, it no longer incorrect to also use the words COM and ActiveX as virtual synonyms. 

Variants get their name because they can provide a wide variety of functions, depending on the circumstances. For instance, they can contain a string, an integer, or, in special cases, a COM object. In other words, the type of variable held in a variant varies from one occasion to the next. That's why they call them variants. (For more information, look up "Variant Types" in the Delphi online help, or else look at the declarations for the structures used with variants at the top of System.pas.) 

CreateOleObject calls a number of internal system wide OLE functions. The end result of these series of calls is that the function returns a COM object to you containing an interface to the object you want to call. In particular, you get back a variant that is wrapped around a COM object called IDispatch. A combination of the built in IDispatch methods, and various Delphi technologies covered briefly later in this paper, allows you to call the methods of the object your requested. 

With all this in mind, let's go back and view the two lines of code that retrieve the Excel object: 

 V := CreateOleObject('Excel.Application');

 V.Visible := True;

The first line of code asks for an object that resides inside Excel called Application. CreateOleObject retrieves an instance of the object in the form of an IDispatch interface encapsulated inside a variant called V. This variant is valuable to you because it allows you to call the methods and properties of the Excel object using a very simple syntax. For instance, you can access the Visible property of the object by simply writing V.Visible := True. 

It would be mistake, however, to assume that the line of code containing the Visible property is doing the same thing as a standard Delphi line of code that looks like this: 

Form1.Visible := True;

Admittedly these two lines look the same, and have exactly the same syntax. But internally, something very different is going on. In particular, if you call the Visible property of a Delphi form object then the property is changed almost instantly. Calling the Visible property of an OLE automation variant sets off a series of internal events that end up resulting in a change to the Visible property of an object inside Excel, but there are many steps that occur along the way. In particular, several methods of IDispatch such as GetIDsOfNames and Invoke must first be called behind the scenes before the call is complete. 

This paper is not designed to cover the mechanisms used in dispatching a call on a variant encapsulated COM object, nor is it necessary for you to understand how it works in order to use this technology. The key point to grasp is merely that things aren't quite as simple as they at first appear. Having said all that, I will now show you how to get into this subject a bit deeper if you so desire, and if you have the source to the VCL on your machine. 

To get started, copy COMObj.pas and COMObj.inc from the Delphi SourceRtlSys directory to the same directory where Excel1 is stored. Now rebuild the project so these local copies of COMObj are linked into your program. Put a breakpoint on the line V.Visible := True, and then run the program. When you get to the breakpoint, press F7 to step into the code. You will find that you are taken immediately to the VarDispInvoke method found in COMObj.pas. From there you will go to GetIDsOfNames, and finally to DispatchInvoke. What is happening here is that the appropriate methods of the IDispatch interface are being called behind the scenes by Delphi in order to "invoke" your call to Excel. 

One of the lessons to be learned from this is that at bottom, there is not such a big difference between the interface technology shown in the second part of this paper and the variant based technology I am discussing here. For instance, IDispatch is an interface, and ultimately this interface must be called for the variant based technology to work. In fact, IDispatch is designed in such a way as to make the variant based technology even more complex than the standard interface technology seen in the second half of this paper. Only Delphi is able to hide that complexity from you, so that you do not need to understand it at all in order to use variants to call automation objects. (Once again, I need to emphasize that I am not giving a full explanation of this technology in this paper. If you want to really understand IDispatch, then you should check out the resources mentioned earlier in this section of the paper.) 

One of the biggest consequences of calling the methods of an object off a variant is that Delphi cannot type check your code at design time. In other words, Delphi does not really know whether or not the Excel Application object has a property called Visible. It is taking you at your word when you claim this is true. In this case, that proves to be the correct thing to do. However, it would also compile without error the following code: 

V.TransferMoney("From := Bill Gates", "To := Charlie Calvert", 100000);

This line of code is certainly intriguing, but the Excel Application object unfortunately does not support it. This means that a program containing it will compile and load without error, but a call to the TransferMoney property at run time will raise an exception. Both Delphi and Excel are able to handle this exception flawlessly, without destabilizing the system in any way. It is nice, however, if you can type check at design time, rather than having to wait to run time to see if all is set up correctly. The interface and dispinterface technologies covered in Part II of this paper shows how to get design time type checking of OLE objects. 

I've spent the last several paragraphs describing IDispatch and variants. This is an important subject, but one that you need not understand in depth in order to use this technology. If all is not clear to you yet, you can still continue without fear. If you are hungry for more details, be patient and I will return to this subject Part II, or else you should follow the links shown earlier in this section. 

After you have created an Excel Application object, you need some way to close it down. You can do this by calling its Quit method: 

if not VarIsEmpty(V) then

  V.Quit;

What this code does is check to make sure that the variant V refers to something, and then it attempts to call the Quit method of the Excel application object. If V is indeed a valid pointer to such an object, then Excel will close. This code is not perfect in all cases, since V could contain a reference to something other than an Excel Application object, thereby allowing VarIsEmpty to return true, even though the call to V.Quit would fail. For instance, I could write: 

V := 10;

After making this call VarIsEmpty would return false, but the call to V.Quit would obviously fail. However, in the Excel1 application, found in Listing 1, V will usually be either empty, or else pointing to a COM object. Therefore the code is reasonably robust. The key point, at any rate, is that you don't want to fail to Quit the Application object or else you can end up cluttering memory with instances of this object. Remember that Excel owns the Application object, and it will not necessarily be removed from memory just because you close your Delphi application. In other words, you should definitely call Application Quit or else repeated calls to Excel from a Delphi application will bog down your machine by draining system resources. 

Creating Excel Automation Objects
Now that you have been introduced to the topic of automating Excel, the next step is to learn something about what it means to create an OLE automation object. 

The call to CreateOleObject returns a COM object called IDispatch housed inside a variant. You can pass a string to CreateOleObject specifying the name of the COM object you wished to retrieve. In this case, I have retrieved the main Excel Automation object, by passing in the string "Excel.Application'. If you are familiar with the registry, you can find this string there, and can trace that reference to the CLSID associated with the LocalServer that returns the object. If you don't know anything about CLSIDs, or about LocalServers, I wouldn't feel too concerned. The point is simply that CreateOleObject returns a COM object of your choice if you pass in the correct string. In particular, it looks up your string in the registry, finds the CLSID associated with the string, looks up the CLSID, and finds the LocalServer associated with that CLSID. The local server will be string pointing at the application that contains the object you want to retrieve. For instance, in this case, on my system, the Local Server string looks like this: 

C:Program FilesMicrosoft OfficeOfficeexcel.exe /automation

This string is copied directly from the REGEDIT.EXE application that ships with all copies of Windows. I found it in HKEY_CLASSES_ROOTCLSID, under the guid listed next to Excel.Application. Guids are 64 byte numbers designed to uniquely identify an object. 

If you want to trace out the details of this operation, and if you have the source to the VCL, you can open up COMObj.pas and find the implementation of CreateOleObject. It consists of a simple call to CoCreateInstance. CoCreateInstance is a Windows API routine that is part of the OLE specification. Its purpose is to retrieve an object from a binary file such as an executable or DLL. 

The strings you pass into CreateOleObject are called ProgIDs. As you just saw, all the ProgIDs valid on your system are listed in the registry under the section HKEY_CLASSES_ROOT. (I cover this subject in more depth in the article on my web site entitled Delphi and COM.) The Delphi documentation is not the place to turn to find the ProgIDs you pass in to the various COM servers available on your system. Instead, you should turn to the documentation for the application you wish to control. For instance, Excel has extensive COM documentation in an online help file that ships with Microsoft Office called VBAXL8.HLP. (Break it down: VBA: Visual Basic for Applications, XL: Excel, 8: Version number.) If you are doing a lot of OLE Automation with Excel then you should add this file to Delphi's tools menu so you can get at it easily. For information on retrieving objects, use the Index feature in the Excel help to look up "OLE programmatic identifiers". 

In the Excel online help, you will find that this spreadsheet application has three main objects you can retrieve using CreateOleObject: 

CreateOleObject('Excel.Application');

CreateOleObject('Excel.Sheet');

CreateOleObject('Excel.Chart');

These strings, and slight variations on these strings, are the only valid parameters to pass to CreateOleObject if you want to talk to Excel via COM. There are many, many more objects inside of Excel. However, these three are the only ones you can retrieve from outside of Excel using the CreateOleObject function. Once you have retrieved one of these objects, you can use it as your access to all the other objects in the Excel hierarchy. Getting at these objects is a bit like unwinding a ball of thread. You first need a handle to the ball of thread, which you get by calling CreateOleObject. Once you have a handle, you can use it to get to all the different objects inside Excel. Just keep pulling at the thread you get back from CreateOleObject and all the rest of the objects will come unraveled. This subject is explained in more depth in the next section. 

Understanding Excel Automation Objects
If you are an experienced Delphi programmer, you may find OLE objects a bit confusing at first. Like standard Pascal objects, they exist inside a hierarchy, but that hierarchy, at least as it is presented to the public, is not based on inheritance. Instead, the main glue that holds the hierarchy together is the fact that you can access one particular object from another particular object. 

For instance, the top member of the Excel hierarchy is called Application. Beneath it is the Workbooks object, and beneath that are the Worksheets and Charts objects: 

1) Application:

  A) Workbooks

    i) Worksheets

    ii) Charts

If you want to get at the Workbooks object, then you can access it from the Application object: 

MyWorkbooks  := Application.Workbooks;

If you want to get at the Worksheets object, then you can access it from the Workbooks object. And so on. In the code shown here, you would declare MyWorkbooks as a variant. In all cases, during this first part of the article, I am using variants to access the underlying Excel objects. Getting an actual interface to a Workbooks interface is covered in the second part of the article. 

If you saw this hierarchy in a Delphi application, you would assume that Workbooks is a descendant of Application, and Worksheets a descendant of Workbooks. That kind of thinking is completely off center when it comes to OLE automation. The standard OOP hierarchy found in C++ and Pascal has nothing to do with OLE Automation. This is a totally different kind of hierarchy intended only to express which objects can be accessed from another object. As you will see in the second part of this paper, it may also be true that there is a valid OOP inheritance based hierarchy simultaneously implemented on this objects. However, that hierarchy is not the main one you focus on when using automation, and in fact, I think it is easiest at first to pretend that it does not exist at all. 

If you want to talk about all the Worksheets and Charts in a Workbook, then you use the Sheets object. When thinking about the Sheets object, you could rewrite the hierarchy shown above as follows: 

1) Application

  A) Workbooks

    i) Sheets

    ii) Worksheets

    iii) Charts

The point is that this hierarchy is meant to denote the order in which you access objects, and as such it has a somewhat more slippery structure than you would fine in a typical inheritance hierarchy. In fact, it seems that you can get at most any object from any one point in the hierarchy, so the actual structure of the hierarchy is a little dependant on your current position inside it. 

You get at the Workbooks object from Application object. You get at the Sheets, Worksheets and Charts objects from the Workbooks object: 

MyCharts := Application.Workbooks[I];

It would be untrue to say that the Application object is synonymous with the binary file Excel.exe, but it does have some things in common with this executable. For instance, the Application object is the most abstracted, the most generalized way that you have of referring to the set of available Excel automation objects. If you open up Excel and have no documents loaded, then you are looking at a visual representation of the Application object. This is not the same thing as the Application object, but it can serve as a metaphor for what the object does. It is analogous to it. It is the highest level container for accessing all of the functionality available from Excel. However, it is so generalized that it can't do much that is useful without help from other objects. But you get at those other objects by starting with the Application object. All this is equally true of Excel.exe. If you open up Excel.exe with no documents in it, then it has little use on its own, but it is still the gateway you would use to access all these documents. 

The WorkBooks object contains a collection of Worksheets and Charts. A Worksheet is just a standard page from a spreadsheet, while a Chart is just a graph. The Sheets object contains both Worksheets and Charts, while the Worksheets and Charts objects contain only Worksheets or Charts. Your job as an Excel automation programmer is to start learning how to make statements like these. In other words, this is the kind of logic that underlies the Excel hierarchy of objects. As an automation programmer your job is to start to figure out how to get at one object from another object, and to understand what each object does. 

Here is another way to think about what you, as an Excel automation programmer, are really trying to do. Most computer users understand how to use Excel. The automation objects discussed in this paper allow you to write code that manipulates Excel just as you would manipulate Excel with a mouse. You probably already know how to open a spreadsheet, enter data, perform calculations, and chart data. You goal as an automation programmer is to find out how to do the same things in code. You just need to know which object refers to which set of tools inside Excel. Figure that out, and figure out how to get at each of these objects given the existence of an Application object, and then you are ready to roll! 

The program shown in Listing 2 provides a summary of the major points made in this section of the paper. Glance over it once, and then read on to find an explanation of how it works. 

Listing 2: The Excel2 program shows how the objects in Excel are arranged hierarchically. 

unit Main;

interface

uses

  Windows, Messages, SysUtils,

  Classes, Graphics, Controls,

  Forms, Dialogs, StdCtrls;

type

  TForm1 = class(TForm)

    Button1: TButton;

    ListBox1: TListBox;

    procedure Button1Click(Sender: TObject);

    procedure FormDestroy(Sender: TObject);

  private

    XLApplication: Variant;

  public

  end;

var

  Form1: TForm1;

implementation

uses

  ComObj;

{$R *.DFM}

procedure TForm1.Button1Click(Sender: TObject);

const

{ XlSheetType }

  xlChart = -4109;

  xlDialogSheet = -4116;

  xlExcel4IntlMacroSheet = 4;

  xlExcel4MacroSheet = 3;

  xlWorksheet = -4167;

{ XlWBATemplate }

  xlWBATChart = -4109;

  xlWBATExcel4IntlMacroSheet = 4;

  xlWBATExcel4MacroSheet = 3;

  xlWBATWorksheet = -4167;

var

  i, j: Integer;

  Sheets: Variant;

begin

  XLApplication := CreateOleObject('Excel.Application');

  XLApplication.Visible := True;

  XLApplication.Workbooks.Add;

  XLApplication.Workbooks.Add(xlWBatChart);

  XLApplication.Workbooks.Add(xlWBatWorkSheet);

  XLApplication.Workbooks[2].Sheets.Add(,,1,xlChart);

  XLApplication.Workbooks[3].Sheets.Add(,,1,xlWorkSheet);

  for i := 1 to XLApplication.Workbooks.Count do begin

    ListBox1.Items.Add('Workbook: ' + XLApplication.Workbooks[i].Name);

    for j := 1 to XLApplication.Workbooks[i].Sheets.Count do

      ListBox1.Items.Add('  Sheet: ' +  

        XLApplication.Workbooks[i].Sheets[j].Name);

  end;

end;

procedure TForm1.FormDestroy(Sender: TObject);

begin

  if not VarIsEmpty(XLApplication) then begin

    XLApplication.DisplayAlerts := False;  // Discard unsaved files....

    XLApplication.Quit;

  end;

end;

end.

This application starts an instance, of Excel, then populates it with three workbooks. One of the workbooks contains a default number of worksheets, a second contains a user defined number of worksheets, and a third contains some workcharts. Over the next few paragraphs I will explain how it all works. 

Take a moment to study the core of the Button1Click method: 

begin

  XLApplication := CreateOleObject('Excel.Application');

  XLApplication.Visible := True;

  XLApplication.Workbooks.Add;

  XLApplication.Workbooks.Add(xlWBatChart);

  XLApplication.Workbooks.Add(xlWBatWorkSheet);

  XLApplication.Workbooks[2].Sheets.Add(,,1,xlChart);

  XLApplication.Workbooks[3].Sheets.Add(,,1,xlWorkSheet);

  for i := 1 to XLApplication.Workbooks.Count do begin

    ListBox1.Items.Add('Workbook: ' + XLApplication.Workbooks[i].Name);

    for j := 1 to XLApplication.Workbooks[i].Sheets.Count do

      ListBox1.Items.Add('  Sheet: ' +  

        XLApplication.Workbooks[i].Sheets[j].Name);

  end;

end;

XLApplication is a variant that contains an instance of IDispatch used for accessing the Excel Application object. As you know, there is a property of Application called Visible. If you set it to True, then Excel will appear on your screen. Once again, this is not the time or place to get into it, but COM objects support the notion of properties. These properties are very different internally from Delphi properties, but behave more or less the same. 

Workbooks is a collection object. It contains a collection of workbooks. This pattern is followed over and over in Excel. The Sheets object contains a collection of sheets. The Worksheets object contains a collection of worksheets. The Charts object contains a collection of charts. Inside Word, the Paragraphs object contains a collection of paragraphs. The Words object contains a collection of words. The Tables object contains a collection of tables. And so on. 

Depending on which automation server you are using, you get at member of a collection through one of four possible syntaxes. Sometimes all syntaxes are available to you, sometimes less: 

MyChart := Charts[1];

MyChart := Charts.Item[1];

MyChart := Charts(1);

MyChart := Charts.Item(1);

You need to be conscious of the difference between a collection object and a normal object. For instance, to understand a Worksheets object, you should look up both Worksheets and Worksheet in the Excel help, to understand the Tables object you should look up both Tables and Table in the Word help. 

Workbooks has a method called Add, which you use to add a workbook to a workbooks collection. COM objects support the idea of variable parameter lists. This means you can simply skip passing in parameters to a method if you want. In this case, if you call Workbooks.Add with no parameters, then you will create a workbook with some predefined number of worksheets in it. The default number is three, but you can change the number from inside of Excel if you so desire. When you are working with interfaces rather than variants, you won't be able to omit parameters. I will explain in the second part of this paper how to work with interfaces in situations such as this one, where not passing in a parameter has a special meaning. 

If you want to create a new Workbook with exactly one Worksheet in it, then you call Add and pass in the constant xlWBatWorksheet. I declare this constant explicitly inside this program. In the next section of this paper I will tell you how to get a complete list of all the Excel and Word constants. 

If you want to create a new workbook with exactly one chart in it, then you call Add and pass in the constant xlWBatChart. 

If you then want to add one worksheet to the second workbook you created, you would write the following code: 

XLApplication.Workbooks[2].Sheets.Add(,,1,xlWorkSheet);

Here is how to create a new chart: 

XLApplication.Workbooks[2].Sheets.Add(,,1,xlChart);

In this case, the Add method of the Sheets object takes four parameters: 

1. Before: A variant containing the sheet before which the new sheet is added. 

2. After : A variant containing the sheet after which the new sheet is added. 

3. Count: The number of sheets to add, with the value defaulting to one. 

4. Type: One of the following constants: xlWorksheet, xlChart, xlExcel4MacroSheet, or xlExcel4IntlMacroSheet. The default value is xlWorksheet. 

The first two parameters specify the location in the workbook where you want the new chart or worksheet to appear. The third parameters states how many sheets you want to add, the fourth specifies the type of sheet you want to add. Here is how the method is declared in the Microsoft docs: 

expression.Add(Before, After, Count, Type);

In the above examples of using Add, I don't care what order the sheets are inserted, so I just omit the first two parameters by simply placing commas in my code where the parameters would be listed. If wanted to state the order, I would write something like this: 

Sheets := Application.Sheets;

Sheets.Add(, Sheets.Item[2], 1, xlChart);

In this case the code still leaves the Before parameter blank, but it references the 2 sheet in the After parameter. 

The for loop at the bottom of the example method iterates through each of the work groups, and then finds the names of each of the sheets available in each workbook, and adds them to list box. In short, the code shows how to retrieve the names of the members of a series of Workbooks, while simultaneously showing how to iterate over all their members. 

Here is how to reference the number of workbooks in the application: 

for i := 1 to XLApplication.Workbooks.Count do begin

And here is how to count the number of sheets in a Workbook: 

for j := 1 to XLApplication.Workbooks[i].Sheets.Count do

Here is how to find the name of a particular Worksheet or Chart in Workbook: 

XLApplication.Workbooks[i].Sheets[j].Name);

If you spend a little while contemplating the Button1Click method, then the logic behind the objects in Microsoft Excel should begin to come clear to you. Of course, there are additional matters to be covered, such as entering data, and creating graphs. But, as you will see, most of that material is relatively straightforward once you understand the way the Excel object hierarchy works. 

One important point to make before closing this section is that it often helps to assign a specific variable to one of the sub-objects in the hierarchy. For instance, in the example shown above I declare a variant named Sheets and set it equal to the Application.Sheets object: 

Sheets := Application.Sheets;

To my mind it is sometimes easier to mentally parse code that is written this way rather than trying to always reference a series of qualified objects such as: 

XLApplication.Workbooks[i].Sheets.Count

Obviously, there is more overhead involved if you use the technique of storing an object reference in a separate variant. However, the technology used to implement OLE Automation on the Excel side is perhaps necessarily not particularly efficient, so you shouldn't balk at using techniques like this if you think they will help you write clear, easy to maintain code. When trying to optimize your code, remember that trips between your application and Excel are very expensive. If you can limit the number of trips you need to make, then you will save clock cycles. But once again, this whole process is innately slow, so it's a bit silly to start fretting over a few lost clock cycles that most users will never even notice. 

As always, you should be particular aware of saving clock cycles when you are inside a loop. A call that takes one second to execute is easy for the user to bear if it occurs once. But put it in a loop, execute it 2000 times, and the user will hate you. A general rule of thumb is that users will happily wait up to two seconds for you to do almost anything. Longer than that and they get impatient. Two seconds is several eons in computer time, so normally you don't have to fret optimization issues. The Delphi team already did all the sweating for you. But when automating Excel or Word, you can get in trouble fairly quickly, so you may need to think about optimization in places where you wouldn't worry about it in a normal Delphi application. Remember that they call Excel automation "Visual Basic for Applications". As you will see later, these Excel classes are actually real objects, so clearly this subject doesn't have much to do with Visual Basic, but the mere presence of the word Basic costs you, by default, thousands of clock cycles! 

Finding the Constants Used in Excel
You can determine all the constants used by Excel by reading its type library. There are at least two simple ways to read a type library. 

1. You can read the type library with a third party tool, such as the OleView application that ships with the Microsoft SDK. 

2. You can ask Delphi to read the library for you, and to translate the information stored in the library into Object Pascal. Obviously, this is the preferred technique. 

I have included the translations of the Excel and Word type libraries with this article. However, if you want to create your own versions of these libraries, then you can select Project | Import Type Library from the Delphi menu, and then select the appropriate type library. A Delphi translation of the type library will be created automatically. (Be sure you are using Delphi 3.01 or later when importing the type library, as some important fixes were made to this technology between Delphi 3.0 and Delphi 3.01.) 

There will be a number of warnings at the top of the type library, but you can ignore them. Most of these warnings have to do with name conflicts. For instance, if Excel uses an Object Pascal keyword in its code, then we need to make an alias for that keyword. For instance, Excel uses End in several places. This is a Delphi keyword, and so we append a space on to the end of it, converting the word End to End_. This puts an end to name conflicts. 

The files you want to import usually have either a TLB or EXE extension. When working with Office 97, however, you want one with an OLB extension. The file to use with Word is MSWORD8.OLB, and the one to use with Excel is EXCEL8.OLB. On my system, I found these entries in the ...Microsoft OfficeOffice directory. 

The Pascal translations of the interfaces to all the objects used in Excel or Word are found in the files created by importing EXCEL8.OLB and MSWORD8.OLB. Throughout this part of the paper, I will ignore these interfaces, and show you how to work directly with variant objects. However, in Part II of the paper I will return to this subject, and show you how to work with interfaces. At that time, I will present a discussion of the relative merits of working with variants and interfaces. 

Storing and Accessing Data in an Excel Worksheet
Throughout the next few sections of this paper I will be working with an example program called Excel3. The source for the main form of this program is shown in Listing 3. Just take a quick look at the code for now, and then read on to get an explanation of how it works. 

Listing 3: The source for the main form of the Excel3 program. 

unit Main;

interface

uses

  Windows, Messages, SysUtils,

  Classes, Graphics, Controls,

  Forms, Dialogs, StdCtrls;

type

  TForm1 = class(TForm)

    Button1: TButton;

    Button2: TButton;

    procedure Button1Click(Sender: TObject);

    procedure FormDestroy(Sender: TObject);

    procedure Button2Click(Sender: TObject);

  private

    XLApp: Variant;

    procedure InsertData;

    procedure ChangeColumns;

    procedure HandleRange;

  public

  end;

var

  Form1: TForm1;

implementation

uses

  ComObj, XLConst;

{$R *.DFM}

procedure TForm1.FormDestroy(Sender: TObject);

begin

  if not VarIsEmpty(XLApp) then begin

    XLApp.DisplayAlerts := False;  // Discard unsaved files....

    XLApp.Quit;

  end;

end;

procedure TForm1.Button1Click(Sender: TObject);

begin

  XLApp:= CreateOleObject('Excel.Application');

  XLApp.Visible := True;

  XLApp.Workbooks.Add(xlWBatWorkSheet);

  XLApp.Workbooks[1].WorkSheets[1].Name := 'Delphi Data';

  InsertData;

  HandleRange;

  ChangeColumns;

end;

procedure TForm1.InsertData;

var

  i: Integer;

  Sheet: Variant;

begin

  Sheet := XLApp.Workbooks[1].WorkSheets['Delphi Data'];

  for i := 1 to 10 do

    Sheet.Cells[i, 1] := i;

  Sheet.Cells[i, 1] := '=Sum(A1:A10)';

end;

procedure TForm1.HandleRange;

var

  Range: Variant;

begin

  Range := 

   XLApp.Workbooks[1].WorkSheets['Delphi Data'].Range['C1:F25'];

  Range.Formula := '=RAND()';

  Range.Columns.Interior.ColorIndex := 3;

  Range.Borders.LineStyle := xlContinuous;

end;

procedure TForm1.ChangeColumns;

var

  ColumnRange: Variant;

begin

  ColumnRange := XLApp.Workbooks[1].WorkSheets['Delphi Data'].Columns;

  ColumnRange.Columns[1].ColumnWidth := 5;

  ColumnRange.Columns[1].Font.Bold := True;

  ColumnRange.Columns[1].Font.Color := clBlue;

end;

procedure TForm1.Button2Click(Sender: TObject);

var

  Sheet: Variant;

  Num: Integer;

begin

  Sheet := XLApp.Workbooks[1].WorkSheets['Delphi Data'];

  Num := Sheet.Range['C1:F25'].Columns.Interior.PatternColor;

  ShowMessage(Format('Value: %x', [Num]));

end;

end.

As I implied above, it is easy to actually insert data in a spreadsheet. In fact, the technique involved is very similar to what you would use putting data into a TStringGrid control in Delphi. 

To get started, open up Excel and create a new spreadsheet: 

procedure TForm1.Button1Click(Sender: TObject);

begin

  XLApp:= CreateOleObject('Excel.Application');

  XLApp.Visible := True;

  XLApp.Workbooks.Add(xlWBatWorkSheet);

  XLApp.Workbooks[1].WorkSheets[1].Name := 'Delphi Data';

  InsertData;

  HandleRange;

  ChangeColumns;

end;

As you can, see, I create a single workbook with one worksheet in it. The code then names the worksheet Delphi Data. The InsertData, HandleRange and ChangeColumns calls are custom Delphi routines which I will now proceed to describe. 

To insert data into the spreadsheet, execute the following function: 

procedure TForm1.InsertData;

var

  i: Integer;

  Sheet: Variant;

begin

  Sheet := XLApp.Workbooks[1].WorkSheets['Delphi Data'];

  for i := 1 to 10 do

    Sheet.Cells[i, 1] := i;

  Sheet.Cells[i, 1] := '=Sum(A1:A10)';

end;

The method starts by retrieving a pointer to the worksheet you want to manipulate. As you know, this pointer is an instance of IDispatch that is stored inside a variant. You don't need to know anything about how IDispatch works in order to call the methods of this object. 

The code proceeds to insert ten integers into the sheet. The Cells property works exactly as you would expect, except Excel puts the Row first and the Column second. 

After inserting the numbers in the worksheet, the final stage is to insert a formula and add up the column of numbers. To do this, you simply insert the formula much as you would if you were in Excel itself. In particular, you store the formula in string, and then insert it into the appropriate cell: 

Sheet.Cells[i, 1] := '=Sum(A1:A10)';

Working with Columns and Range Attributes
Sometimes you might want to perform an operation on a range of data in the spreadsheet. To do this, you use the Excel Range object: 

Sheet.Range['C1:F25'].Formula := '=RAND()';

This code could be inserted into the bottom of the InsertData method. It fills all the cells between C1 and F25 with random numbers between 0 and 1. 

One of the key objects in both Excel and Word is the Range object. It allows you to work with a range of cells or columns at one time. In either Word or Excel, you generally enter or read data by using the Range object. In short, if you want to insert text into a Word document, then you will generally use the Range object! 

BEGIN NOTE: 

With Word, it is also possible to insert data via a considerably simpler method. For instance, the following procedure will enter data at the current insertion point into a currently open document in Microsoft Word: 

procedure TForm1.Button1Click(Sender: TObject);

var

  V: Variant;

begin

  V := GetActiveOleObject('Word.Basic');

  V.Insert('Sam');

end;

In this case, I have chosen not to open a new version of Word, but instead call GetActiveOleObject to get a handle to a document in an instance of Word that is already running. This kind of technology is very easy to use, and is perfect for some projects. It does not, however, have the power of the technology I am showing you in this paper. 

END NOTE 

To access a Range object in Excel, simply specify the Range with which you want to work: 

Range := Sheet.Range['C1:F25'];

In this case the code defines a range from cell C1 to cell F25. Any operations performed on the returned Range object will affect all the cells in that range. 

Here is a simple function showing how to change the values and appearance of a range of cells: 

procedure TForm1.HandleRange;

var

  Range: Variant;

begin

  Range := 

   XLApp.Workbooks[1].WorkSheets['Delphi Data'].Range['C1:F25'];

  Range.Formula := '=RAND()';

  Range.Columns.Interior.ColorIndex := 3;

  Range.Borders.LineStyle := xlContinuous;

end;

The first line of code returns a pointer to the range you want to manipulate. The second line fills all the values in the range C1:F25 with random numbers between 0 and 1, as explained earlier. 

The third line of code changes the color of the entire block of cells to red. You can use the Excel online help to see the values in the ColorIndex, but the first few default values are as follows: black, white, red, green, blue, yellow, purple, cyan. Red is the third item in the list, so setting the ColorIndex to 3 makes the selected range of cells Red. 

At the same time that you change the color, you also loose your borders. This is a peculiarity of Excel, and it can be worked around by resetting the LineStyle of the selected cells as shown in the last line of code in the procedure. Once again, when you are working with constants like this, you can find them in the XLCONST.pas or EXCELTLB.pas files included with the example programs accompanying this article, or can retrieve them from the type library as explained earlier. 

For those who are interested, here is a line of code that changes the background of a range: 

Range.Columns.Interior.Pattern := xlPatternCrissCross;

The following function changes the width and font of a column: 

procedure TForm1.ChangeColumns;

var

  ColumnRange: Variant;

begin

  ColumnRange := XLApp.Workbooks[1].WorkSheets['Delphi Data'].Columns;

  ColumnRange.Columns[1].ColumnWidth := 5;

  ColumnRange.Columns[1].Font.Bold := True;

  ColumnRange.Columns[1].Font.Color := clBlue;

end;

As you can see, when you want to work with the columns in a worksheet, you can access them from a Range object. In particular, the Range object contains a collection of columns that you can access using array notation. 

To change the width of a column, use the ColumnWidth property, and to change the Font, use the Font property. Going into much more detail would be pointless, as this code is easy to write. 

Creating a Chart
It is just as easy to create and work with a Chart as it is to do everything else in Excel automation. In the example shown in this section, refer to the online program called Excel4.dpr. The listing for this program is shown in Listing 4. I include the listing here so that you can take a quick glance through it, and then refer back to it during the discussion of its inner workings that follows this listing. In other words, I don't expect you to understand the program completely at a single glance, but will instead spend the remainder of this section discussing it in some depth. 

Listing 4: The Excel4 programs shows how to work with charts. 

unit Main;

{

  Main.pas

  Copyright (c) 1997 by Charlie Calvert

  Creating data and a chart in Excel and copying both to Word.

}

interface

uses

  Windows, Messages, SysUtils,

  Classes, Graphics, Controls,

  Forms, Dialogs, StdCtrls;

type

  TForm1 = class(TForm)

    Button1: TButton;

    procedure Button1Click(Sender: TObject);

    procedure FormDestroy(Sender: TObject);

  private

    XLApp: Variant;

    WordApp: Variant;

  public

    procedure HandleData;

    procedure ChartData;

    procedure CopyData;

    procedure CopyChartToWord;

    procedure CopyCellsToWord;

    procedure MailDocument;

  end;

var

  Form1: TForm1;

implementation

uses

  ComObj, XLConst, WordConst,

  ActiveX;

{$R *.DFM}

procedure TForm1.Button1Click(Sender: TObject);

begin

  XLApp := CreateOleObject('Excel.Application');

  XLApp.Visible := True;

  XLApp.Workbooks.Add[XLWBatWorksheet];

  XLApp.Workbooks[1].Worksheets[1].Name := 'Delphi Data';

  HandleData;

  ChartData;

  CopyData;

  MailDocument;

end;

procedure TForm1.HandleData;

var

  Sheet: Variant;

  i: Integer;

begin

  Sheet := XLApp.Workbooks[1].Worksheets['Delphi Data'];

  for i := 1 to 10 do

    Sheet.Cells[i, 1] := i;

end;

procedure TForm1.ChartData;

var

  ARange: Variant;

  Sheets: Variant;

begin

  XLApp.Workbooks[1].Sheets.Add(,,1,xlChart);

  Sheets := XLApp.Sheets;

  ARange := Sheets.Item['Delphi Data'].Range['A1:A10'];

  Sheets.Item['Chart1'].SeriesCollection.Item[1].Values := ARange;

  Sheets.Item['Chart1'].ChartType := xl3DPie;

  Sheets.Item['Chart1'].SeriesCollection.Item[1].HasDataLabels := True;

  XLApp.Workbooks[1].Sheets.Add(,,1,xlChart);

  Sheets.Item['Chart2'].SeriesCollection.Item[1].Values := ARange;

  Sheets.Item['Chart2'].SeriesCollection.Add(ARange);

  Sheets.Item['Chart2'].SeriesCollection.NewSeries;

  Sheets.Item['Chart2'].SeriesCollection.Item[3].Values :=

    VarArrayOf([1,2,3,4,5, 6,7,8,9,10]);

  Sheets.Item['Chart2'].ChartType := xl3DColumn;

end;

procedure TForm1.CopyData;

var

  Sheets: Variant;

begin

  SetFocus;

  Sheets := XLApp.Sheets;

  Sheets.Item['Delphi Data'].Activate;

  Sheets.Item['Delphi Data'].Range['A1:A10'].Select;

  Sheets.Item['Delphi Data'].UsedRange.Copy;

  CopyCellsToWord;

  Sheets.Item['Chart1'].Select;

  XLApp.Selection.Copy;

  CopyChartToWord;

end;

procedure TForm1.CopyChartToWord;

var

  Range: Variant;

  i, NumPars: Integer;

begin

  NumPars := WordApp.Documents.Item(1).Paragraphs.Count;

  Range := WordApp.Documents.Item(1).Range(

    WordApp.Documents.Item(1).Paragraphs.Item(NumPars).Range.Start,

    WordApp.Documents.Item(1).Paragraphs.Item(NumPars).Range.End);

  Range.Text := 'This is graph: ';

  for i := 1 to 3 do WordApp.Documents.Item(1).Paragraphs.Add;

  Range := WordApp.Documents.Item(1).Range(

    WordApp.Documents.Item(1).Paragraphs.Item(NumPars + 1).Range.Start,

    WordApp.Documents.Item(1).Paragraphs.Item(NumPars + 1).Range.End);

  Range.PasteSpecial(,,,,wdPasteOleObject);  

end;

procedure TForm1.CopyCellsToWord;

var

  Range: Variant;

  i: Integer;

begin

  WordApp := CreateOleObject('Word.Application');

  WordApp.Visible := True;

  WordApp.Documents.Add;

  Range := WordApp.Documents.Item(1).Range;

  Range.Text := 'This is a column from a spreadsheet: ';

  for i := 1 to 3 do WordApp.Documents.Item(1).Paragraphs.Add;

  Range := 

    WordApp.Documents.Item(1).Range(WordApp.Documents.Item(1).      

      Paragraphs.Item(3).Range.Start);

  Range.Paste;

  for i := 1 to 3 do WordApp.Documents.Item(1).Paragraphs.Add;

end;

procedure TForm1.FormDestroy(Sender: TObject);

begin

  if not VarIsEmpty(XLApp) then begin

    XLApp.DisplayAlerts := False;  // Discard unsaved files....

    XLApp.Quit;

  end;

  if not VarIsEmpty(WordApp)then begin

    WordApp.Documents.Item(1).Close(wdDoNotSaveChanges);

    WordApp.Quit;

  end;

end;

procedure TForm1.MailDocument;

begin

  WordApp.Documents.Item(1).SaveAs('c:foo.doc');

  WordApp.Options.SendMailAttach := True;

  WordApp.Documents.Item(1).SendMail;

end;

end.

This code pops up a copy of Excel, inserts some data into it, creates two graphs of the data, then pops up a copy of Word, copies the cells from the worksheet to a new Word document, and then copies one of the charts into the same document. When you are through, you have a word document containing some spreadsheet cells with Delphi data in them, and below these cells, a graph. You may not see the graph at first when looking at your copy of Word. To find the graph, scroll the document down a bit. By default, there is a fairly large margin at the top of a graph, so you may need to scroll down further than you think. After creating the Word document, the user has a chance to mail it via Microsoft mail. 

The Button1Click method drives the entire application: 

procedure TForm1.Button1Click(Sender: TObject);

begin

  XLApp := CreateOleObject('Excel.Application');

  XLApp.Visible := True;

  XLApp.Workbooks.Add[XLWBatWorksheet];

  XLApp.Workbooks[1].Worksheets[1].Name := 'Delphi Data';

  HandleData;

  ChartData;

  CopyData;

  MailDocument;

end;

Its starts by creating an Excel Application object, then sets the Visible property of the object to true, and adds a new workbook and stuffs a single worksheet into it. The Delphi application then calls my custom HandleData method to insert data into the spreadsheet: 

procedure TForm1.HandleData;

var

  Sheet: Variant;

  i: Integer;

begin

  Sheet := XLApp.Workbooks[1].Worksheets['Delphi Data'];

  for i := 1 to 10 do

    Sheet.Cells[i, 1] := i;

end;

This method was explained earlier in this article, so I won't cover it again. 

Now that you have a worksheet, and some data, then next step is to create a graph. The following procedure from the Excel4 program should get you started working with charts: 

procedure TForm1.ChartData;

var

  ARange: Variant;

  Sheets: Variant;

begin

  XLApp.Workbooks[1].Sheets.Add(,,1,xlChart);

  Sheets := XLApp.Sheets;

  ARange := Sheets.Item['Delphi Data'].Range['A1:A10'];

  Sheets.Item['Chart1'].SeriesCollection.Item[1].Values := ARange;

  Sheets.Item['Chart1'].ChartType := xl3DPie;

  Sheets.Item['Chart1'].SeriesCollection.Item[1].HasDataLabels := True;

  XLApp.Workbooks[1].Sheets.Add(,,1,xlChart);

  Sheets.Item['Chart2'].SeriesCollection.Item[1].Values := ARange;

  Sheets.Item['Chart2'].SeriesCollection.Add(ARange);

  Sheets.Item['Chart2'].SeriesCollection.NewSeries;

  Sheets.Item['Chart2'].SeriesCollection.Item[3].Values :=

    VarArrayOf([1,2,3,4,5, 6,7,8,9,10]);

  Sheets.Item['Chart2'].ChartType := xl3DColumn;

end;

This function creates two different charts. I've arranged things this way so you can get a look at some of the different techniques needed to create charts. 

The code starts by adding a single chart to a Sheets object in a workbook: 

XLApp.Workbooks[1].Sheets.Add(,,1,xlChart);

As you can see, I just ignore the first two parameters, then explicitly state that I want to insert 1 sheet, and define its type as xlChart. The Add method was described in more depth earlier in this article. 

A lot of the trick to working with Excel is to find the right object to work with. In my opinion, the Sheets object provides a simple and convenient way to create a chart, but it is not the only way of doing so. Remember that the Sheets object contains both the Worksheets and Charts objects inside a Workbook, so you can use it to add either worksheets or charts. 

You should further understand that I am talking about adding Charts to a Sheets object, which is different from adding ChartObjects to a worksheet. In other words, you can insert a graph into a worksheet, but that is a different operation from the one I show here. The key to embedding a chart into a work sheet is the Excel ChartObjects collection, which is not discussed further in this article. 

Once the chart has been created, the code then finds a range of data in the sheet to work on. In this particular example, the range is the same as produced back in the Excel 3 application, when I inserted 10 numbers into the A column, and then supplied a formula to add them up. In particular, note that I create a range object, then set the SeriesCollection of a Chart object to this range: 

  ARange := Sheets.Item['Delphi Data'].Range['A1:A10'];

  Sheets.Item['Chart1'].SeriesCollection.Item[1].Values := ARange;

That is all you need to do to graph a range of data. As I will explain in a moment, you may want to manipulate the chart further, but just doing what I have done here is enough to start charting data. 

It is probably worthwhile stepping back and looking at the SeriesCollection object and see what it represents. To get started, you needed to understand that a Series is simply a range of data that you want to graph. A SeriesCollection is a collection of ranges of data, that is, it is a collection of Series. For instance, if you had the values 1, 2, 3 in three cells in a spread sheet, then that would represent a range of three numbers which could be placed in a Series. By default, the graph of that series might look something like this, where each dash represents one unit in the Series 1, 2, 3: 

- 

-- 

--- 

If you had several Series together in one place, then that would be a SeriesCollection. 

To see a SeriesCollection on the Excel side, load an Excel chart, right click on it, select the Source Data item from the menu, and turn to the Series page, as shown in Figure 1: 

Figure 1: Showing a series inside of Excel. 

This is the series created for the first graph made by the Excel4 program. As you can see, the series is described with a single cryptic line: 

='Delphi Data'!$A$1:$A$10

If you think about it for a second, you can see how this line corresponds to the code you wrote in Delphi. To help you see the relationship, I will write a single line of pseudo-code designed to highlight the relationship: 

  SeriesCollection.Item[1].Values := 

    Sheets.Item['Delphi Data'].Range[A1:A10];

This line will not compile, but it contains the essence of what happens in the two lines of Delphi code which creates the Series shown from the SeriesCollection. It is similar to the data seen on the Excel side. 

When you create a new chart, there is one Series made for you automatically. By default, it charts whatever value is in cell A1 of a particular Worksheet. In this case, I have changed that series to point to a new range of data. In other words, I have changed the "Values" associated with the Series. As you will see in one moment, you can add additional series if you so desire. 

After creating the chart, the code defines it further by stating its type: 

Sheets.Item['Chart1'].ChartType := xl3DPie;

And then goes on to specify that the chart has a series of data labels: 

Sheets.Item['Chart1'].SeriesCollection.Item[1].HasDataLabels := True;

Its time now to look at the second chart created by the ChartData method: 

  XLApp.Workbooks[1].Sheets.Add(,,1,xlChart);

  Sheets.Item['Chart2'].SeriesCollection.Item[1].Values := ARange;

  Sheets.Item['Chart2'].SeriesCollection.Add(ARange);

  Sheets.Item['Chart2'].SeriesCollection.NewSeries;

  Sheets.Item['Chart2'].SeriesCollection.Item[3].Values :=

    VarArrayOf([1,2,3,4,5,6,7,8,9,10]);

This chart graphs a SeriesCollection that contains not one, but three Series. The first and second Series are identical to the Series graphed by the first chart, but the third Series is slightly different, in that its values come not from an Excel worksheet, but from a range of data directly specified inside Delphi. 

Take a moment to consider what is happening here. The first set of data graphed is specified exactly as in the previous example: 

Sheets.Item['Chart2'].SeriesCollection.Item[1].Values := ARange;

Adding a new range specifies the next Series: 

Sheets.Item['Chart2'].SeriesCollection.Add(ARange);

Finally, the code creates a new Series with no particular data: 

Sheets.Item['Chart2'].SeriesCollection.NewSeries;

The program than creates a variant array containing the values you want to chart in this third Series: 

  Sheets.Item['Chart2'].SeriesCollection.Item[3].Values :=

    VarArrayOf([1,2,3,4,5,6,7,8,9,10]);

This is not the time nor place to discuss variant arrays, but the subject is treated in Chapter 25 of the Delphi User's Guide. The series created by this code is shown on the Excel side in Figure 2. 

Figure 2: A variant array created in Delphi as it is depicted inside Excel. 

In this case, I need not change the type of graph, since the default column chart does the job adequately. I could, however, add the following line of code to create a new effect: 

Sheets.Item['Chart2'].ChartType := xl3DColumn;

You are at last ready to copy the worksheet and two charts that you have created from Excel into Word. This is a relatively complex operation, so I will start a new section in this article to discuss it. 

Copying Data from Excel to Word
The process of copying data from Excel to Word has two parts. The first part involves copying the data to the clipboard, and the second part involves pasting the data into the Word document. In other words, you have to have both Excel and Word open to make this work. Furthermore, the tricky part is not so much copying the data from Excel, but inserting it correctly into Word. 

Below this paragraph is the procedure which copies the data from Excel to the clipboard. Note that I have created two methods called CopyCellsToWord and CopyChartToWord. These methods handle the Word side of the process, allowing me to isolate the Excel code in on method: 

procedure TForm1.CopyData;

var

  Sheets: Variant;

begin

  SetFocus;

  Sheets := XLApp.Sheets;

  Sheets.Item['Delphi Data'].Activate;

  Sheets.Item['Delphi Data'].Range['A1:A10'].Select;

  Sheets.Item['Delphi Data'].UsedRange.Copy;

  CopyCellsToWord;

  Sheets.Item['Chart1'].Select;

  XLApp.Selection.Copy;

  CopyChartToWord;

end;

To copy data from a range in a worksheet to the clipboard I first retrieve a Sheets object, then do the following: 

  Sheets.Item['Delphi Data'].Activate;

  Sheets.Item['Delphi Data'].Range['A1:A10'].Select;

  Sheets.Item['Delphi Data'].UsedRange.Copy;

I first Activate the worksheet, then select a range of data in it, and finally copy the data to memory. In this process, I essentially mirror the actions I would take were I doing this manually. In other words, I first "click" on the work sheet I want to use, that is, I activate it. I then select a range of data from it, and finally I "press Ctrl-C" to copy it to the clipboard. Of course, I'm not really doing these things, but I'm executing in code the steps necessary to duplicate these actions as follows: 

	Call Activate 
	Click on page with mouse 

	Call Select 
	Select data with the mouse 

	Call Copy 
	Press Crtl-C or pull down the Edit menu and choose Copy 


After the program copies a range of cells to memory, the next step is to copy the cells to Word. In this explanation, however, I will temporarily pass over the act of copying the data to word, and instead show you how to copy the Chart to memory. Note however, that you obviously must do these things one at a time, since the clipboard can only hold one object in memory at a time. In short, you can't copy both the worksheet and the chart to two separate places in the clipboard, then copy them both to Word in one motion. The problem, of course, is that the clipboard has only one area available in memory. 

Here is how to copy a chart to the clipboard: 

Sheets.Item['Chart1'].Select;

XLApp.Selection.Copy;

This code first selects Chart1, then copies it to the clipboard. Again, I am mirroring the actions I would take where I doing this all manually. That is, I first select the object, then "press Ctrl-C" to copy it. Once again, I don't explicitly press Crtl-C, but instead perform the steps in code that duplicate this action. 

Automation Inside Word
In the next few paragraphs I will be discussing the following procedure, which gets you up and running with Automation in Word: 

procedure TForm1.CopyCellsToWord;

var

  Range: Variant;

  i: Integer;

begin

  WordApp := CreateOleObject('Word.Application');

  WordApp.Visible := True;

  WordApp.Documents.Add;

  Range := WordApp.Documents.Item(1).Range;

  Range.Text := 'This is a column from a spreadsheet: ';

  for i := 1 to 3 do WordApp.Documents.Item(1).Paragraphs.Add;

  Range := 

    WordApp.Documents.Item(1).Range(WordApp.Documents.Item(1).       

    Paragraphs.Item(3).Range.Start);

  Range.Paste;

  for i := 1 to 3 do WordApp.Documents.Item(1).Paragraphs.Add;

end;

To get started in Word, you just follow more or less the same steps you would in Excel: 

  WordApp := CreateOleObject('Word.Application');

  WordApp.Visible := True;

  WordApp.Documents.Add;

This code creates a Word Application object, sets the Visible property of the object to True, and adds a single document to it. 

To add text to the document, you can execute the following code: 

  Range := WordApp.Documents.Item(1).Range;

  Range.Text := 'This is a column from a spreadsheet: ';

In this case the code retrieves a Range object representing the entire document, which of course starts out completely empty. To start to place text in the document, you can use the Text property of the Range. 

You could simply paste the data from Excel directly in your document. However, you want to be able to have some control over the location where the cells are placed. To do this, you need some white space in the document, that is, you need a series of carriage returns through which you can iterate: 

for i := 1 to 3 do WordApp.Documents.Item(1).Paragraphs.Add;

You can now use the Goto method of the Range or Document object to move back and forth across this range of paragraphs. Or, if you want, you can select a new Range, and then Paste your Excel data into that Range. In my experience, this second method is a simpler method of moving through a document. Here is the code for selecting a Range covering the third paragraph of a document: 

Range := 

    WordApp.Documents.Item(1).Range(WordApp.Documents.Item(1).       

    Paragraphs.Item(3).Range.Start);

Please note that this is one single line of code. I wrap it here because I must in order to fit it in a word processing or HTML document. In your code, however, you want to type it in on one single line. This code states that I want to define a Range on the third paragraph of the document. I explicitly state that the Range starts at the beginning of the paragraph, but I do not define the end of the Range. In a moment I will show you how to also specify the end of a Range. 

I can now Paste in the Excel code with a single easy to write line: 

Range.Paste;

After pasting in this Range of Cells, I find that several new Paragraphs have been added to my document. There is no specific way for me to be sure how many, since the number of cells I paste in may vary with different versions of my program. So when get ready to paste in the Chart from Excel, I begin by asking how many paragraphs are currently in the document: 

procedure TForm1.CopyChartToWord;

var

  Range: Variant;

  i, NumPars: Integer;

begin

  NumPars := WordApp.Documents.Item(1).Paragraphs.Count;

  Range := WordApp.Documents.Item(1).Range(

    WordApp.Documents.Item(1).Paragraphs.Item(NumPars).Range.Start,

    WordApp.Documents.Item(1).Paragraphs.Item(NumPars).Range.End);

  Range.Text := 'This is graph: ';

  for i := 1 to 3 do WordApp.Documents.Item(1).Paragraphs.Add;

  Range := WordApp.Documents.Item(1).Range(

    WordApp.Documents.Item(1).Paragraphs.Item(NumPars + 2).Range.Start,

    WordApp.Documents.Item(1).Paragraphs.Item(NumPars + 2).Range.End);

  Range.PasteSpecial(,,,,wdPasteOleObject);  

end;

I return the paragraph count in the variable NumPars. I then create a domain that ranges over the last paragraph of the document. In other words, I count the paragraphs in the document, and then say I want to establish a Range on the last paragraph. Once again, this is one way to position yourself in the document: 

Range := WordApp.Documents.Item(1).Range(

WordApp.Documents.Item(1).Paragraphs.Item(NumPars).Range.Start,

WordApp.Documents.Item(1).Paragraphs.Item(NumPars).Range.End);

Once I've located myself in the proper position, the next step is to enter a single descriptive line of text, followed by a few additional paragraphs: 

Range.Text := 'This is graph: ';

for i := 1 to 3 do WordApp.Documents.Item(1).Paragraphs.Add;

I then once again position myself on the last paragraph in the document: 

Range := WordApp.Documents.Item(1).Range(

  WordApp.Documents.Item(1).Paragraphs.Item(NumPars + 1).Range.Start,

  WordApp.Documents.Item(1).Paragraphs.Item(NumPars + 1).Range.End);

Notice that when creating this Range I explicitly state that it "ranges" from the beginning of the paragraph to the end. In an earlier example, I filled in the first part of the range, but left the second part open. In this case, either method would work, but I show you both so you can see various examples of how the syntax works. 

In that same spirit, I use a slightly different technique when pasting in the Chart data: 

Range.PasteSpecial(,,,,wdPasteOleObject);  

In this case I call the PasteSpecial method, and ask to insert an OLE object. This insures that you can edit the document in place by double clicking on it. The PasteSpecial method takes a wide range of parameters, but I won't discuss them here, since that subject is fairly tangential to this paper. For more details, look up PasteSpecial in the Word Visual Basic help. Remember that this help file is not installed by default, so you may need to run the Word install and explicitly ask for the file, or else copy it off the CD manually. 

Mailing a Document
The Documents object also has Save, SaveAs, and Open methods you can use when opening or saving a document. In fact, there are many methods and properties associated with most of the objects discussed in this article. The only way to get to know them all is to open up the Word or Excel help and start browsing through them. This article is meant to give you a conceptual overview of how to use Word and Excel automation objects. It is not a complete examination of the subject. 

Here is how to save the current Word document: 

WordApp.Documents.Item(1).SaveAs('c:foo.doc');

The following code allows you to send a mail message: 

procedure TForm1.MailDocument;

begin

  WordApp.Documents.Item(1).SaveAs('c:foo.doc');

  WordApp.Options.SendMailAttach := True;

  WordApp.Documents.Item(1).SendMail;

end;

If you have Microsoft Mail set up on your computer, you can send a document to another user directly from Excel. The code for doing this is shown above. This code will automatically pop up the mail services, allowing you to pick a user and send the document. The document you created will automatically be attached to your message. If you don't have mail set up on your machine, then this code obviously won't work. 

Summary of Part I
That is all I'm going to say on the basic steps involved with automating Excel and Word from Delphi. This is obviously a large subject, and much more could be said about it. However, the information you have seen here should open up the topic sufficiently to allow you to perform most tasks. 

By now it should be obvious to you that the automation objects in Excel and Word are extremely powerful. Were you creating an actual word processor, you couldn't ask for much more in terms of functionality, but of course you would look for a little better performance! 

The next installment of this paper, found in Part II, will cover interfaces and dispinterfaces. These are extremely important topics, and indeed your education in automating Excel or Word is definitely not complete without an understanding of these important topics. In fact, you will find that in Part II, I rewrite all the code shown you so far, giving you a second, somewhat more complicated, but considerably more powerful set of tools. So far I've shown you code that duplicates more or less what you can do in Visual Basic. The second part of the article shows how you can leverage the power of Delphi to take full advantage of automation. 

